

# le jeu d'awele

par Jérémie Cosmao et Pierre Gélard, élèves de 1° S au lycée Corneille de Rouen (76)

enseignant : Jean Toromanoff

chercheur : Claude Delacherie

Compte-rendu de l'exposé par les parrains du groupe : **CLG V. Hugo**


— Pour trouver la stratégie du quarto, il faut manipuler des "arbres", et il faut surtout prendre en compte les critères de chacun des pions (troués, grands, ...). Pour l'Awélé, il faut aussi manipuler des arbres. Pour les deux, c'est souvent le joueur 1 qui gagne.

— Le jeu se joue dans :




Il y a dans chaque case 4 billes. Le 1<sup>er</sup> joueur prend les billes dans l'une de ses cases et les distribue dans les autres cases (dans le sens inverse des aiguilles d'une montre), une dans chaque case. Si à la fin de sa distribution, il tombe sur une de ses cases, il continue (si sa case est vide, il prend les billes en face et il passe la main) ; si il finit sur une case adverse, c'est au tour de l'autre. Le gagnant est celui qui a gagné plus de la moitié des billes.

Pour simplifier, on réduit le "terrain"



On s'aperçoit alors, après avoir construit un arbre de solutions, qu'il existe des boucles : un même joueur joue indéfiniment car il fait un cycle.


par exemple :



On finit par la case ayant le motif \*.

## Règles du jeu


Cette variante du jeu de l'Awéle (version ghanéenne) comporte deux rangées de six cases contenant chacune quatre billes au début du jeu.



Au début, 4 billes pour chaque case et 6 cases par joueur.

Ce jeu se joue à deux : chacun possède un camp (composé d'une ligne de 6 cases). Le but du jeu est de gagner le maximum de billes. Le jeu s'arrête lorsqu'un joueur a gagné plus de la moitié des billes du jeu.

Le premier joueur sélectionne une des cases de son camp. Il en prend les billes et les dépose une à une dans les cases suivantes (sens contraire des aiguilles d'une montre).




Il prend ensuite les billes de la case où il a déposé la dernière, y compris celle-ci, et les distribue comme précédemment.

Il recommence jusqu'à ce que la dernière bille qu'il dépose tombe dans une case vide.


Deux cas se présentent alors :

- Si cette case fait partie du camp adverse, il ne gagne pas de billes et c'est au joueur adverse de jouer.
- Si cette case fait partie de son camp, il prend les billes de la case en face, puis c'est au joueur adverse de jouer.


Le joueur 1 a commencé, choisi la case 3, s'est arrêté à la case 7, qui n'était pas vide. Il reprend les billes et tourne à nouveau jusqu'à ce qu'il tombe sur une case vide ...  
Ce qui donne :




case 12



case 5




case 11



case 5

Et voilà ! Le joueur 1 s'est arrêté.  
Or il est justement dans son camp.  
Donc il prend les billes de la case face à lui : IL GAGNE 6 BILLES et c'est au joueur 2 de jouer ...

***A qui fera le plus de billes !***



### Les "Boucles Infinies"

Une boucle infinie est un cas très rare où le choix d'un joueur entraîne l'impossibilité de continuer la partie.

Lors de notre étude du jeu, nous avons d'abord limité, pour des raisons de simplicité, notre recherche à l'Awele à 4 cases. Les règles sont les mêmes, mais le camp des joueurs ne comporte que 2 cases.

Tout de suite, un exemple (Awele à 4 cases) :  
Le joueur du bas a choisi la case en bas, à droite.


2 0 3 1 0 2 1 0 2 1 0 1 1 2 1 3 2 0  
1 2 1 0 2 1 3 1 0 2 1 3 2 0 0 1 1 2  
→ → → → → → → → On est revenu au tableau de départ !!! La partie ne finit plus.

La période de cette boucle est 8. Pour l'Awele à 12 cases, certaines boucles ont des périodes très, très longues.

ex : Si le joueur 2 joue la case notée "x" ... (ce tableau existe vraiment),

	x				
11	2	1	4	0	6
1	6	3	11	1	0

la période est ... 50635 !!!!!




**CONCLUSION :** Les boucles infinies ne sont que des "défauts". Heureusement, elles sont extrêmement rares. Environ 1 chance sur 1296. [NDLR : comment sait-on que ce n'est pas "environ 1 chance sur 1297" ? parce que l'infini vaut 5000 ? (programme, page 124)]

### ***Le programme (en BASIC)***

Le but de notre recherche était de trouver un moyen de gagner. N'ayant pas trouvé de stratégies très probantes, nous avons décidé de résoudre le problème par l'envisagement de toutes les possibilités de jeu (grâce à l'ordinateur) sur 4 niveaux. A chaque coup, l'ordinateur envisage donc  $6^4 = 1296$  possibilités.

La partie "intéressante" du programme est celle qui concerne la détermination du meilleur coup possible pour l'ordinateur.

Imaginons un tableau d'Awele à 4 cases. Chaque joueur a donc 2 possibilités à chaque fois qu'il doit jouer. (Les scores de la fin sont fictifs. Ils s'obtiennent en additionnant le nombre de billes gagnées par le joueur 1 et en soustrayant le nombre de billes gagnées par le joueur 2.)



### ***Explications :***

Au niveau 3, ce serait au joueur 1 de jouer. Il choisirait donc les scores qui lui seraient favorables, c'est à dire les scores les plus élevés (notés : +). Au niveau 2, ce serait au joueur 2 de jouer. Il choisirait les scores les plus bas (notés : -).

Le joueur 1 sait donc, que sur 3 niveaux, l'espérance de gain s'il joue la deuxième case est 1 bille. C'est mieux que pour la première (aucune bille). Il choisit donc la deuxième case.

Ce qui est valable pour l'awele à 4 cases sur 3 niveaux l'est aussi pour l'awele à 12 cases sur 4 niveaux. L'arbre de choix est simplement plus important ... Les branches impossibles (cases vides ou boucles infinies) sont écartées du choix de l'ordinateur par un artifice informatique.

### ***conclusion***

Le programme marche très bien. Il a gagné 100% des parties contre le hasard. Personnellement, je n'ai pas encore réussi à le battre.

[NDLR : une recherche systématique des cycles sur les jeux à semis répétés constituerait une recherche mathématique intéressante.]

**annexe : programme en Basic**

```

10 DEFINT A-Z:DEFDBL T:T0=TIMER
20 DIM Z(6^4,12),DE(12),A(12),CAS(13),BAB(
    100,12):CLS
30 GOSUB 1500:CLS
40 RANDOMIZE TIMER
50 FOR I=1 TO 12:CAS(I)=4:NEXT I
60 PRINT "(1)-> Vous jouez contre l'ordina
    teur"
70 PRINT "(2)-> 2 joueurs"
80 INPUT "Votre choix: ",CHO
90 IF CHO<>1 AND CHO<>2 THEN PRINT "On a dit
    1 ou 2 !!":GOTO 60
100 INPUT "Qui commence ? (1 ou 2)";JO
110 PRINT "joueur"JO"commence."
120 T1=TIMER
130 IF JO=1 AND CHO=1 THEN GOSUB 870:GOTO
    190
140 IF JO=1 AND CHO=2 THEN GOSUB 820:GOTO
    190
150 IF JO=2 THEN GOSUB 760:GOTO 190
160 PRINT "Comment ca, le joueur"JO"? J'ai
    dit de choisir entre 1 et 2 !!"
170 GOTO 100
180 '-----
190 B=CAS(CH):CAS(CH)=0
200 FOR I=1 TO B
210     CH=CH+1
220     IF CH=13 THEN CH=1
230     CAS(CH)=CAS(CH)+1
240 NEXT I:SW3=SW3+1
250 IF SW3<100 THEN FOR I=1 TO
    12:BAB(SW3,I)=CAS(I):NEXT I
260 IF SW3=100 THEN FOR I=1 TO
    12:BAB(SW3,I)=99:NEXT I
270 IF SW3=5000 THEN PRINT"Vous etes tombe
    sur une boucle infinie":GOTO 620
280 IF CAS(CH)>1 THEN GOTO 190
290 IF SW3<100 THEN FOR I=1 TO
    12:BAB(SW3+1,I)=99:NEXT I
300 SW3=0:NOMB=NOMB+1
310 IF JO=1 AND CH<7 THEN S(1)=S(1)+CAS(
    CH+(6-CH)*2+1):CAS(CH+(6-CH)*2+1)=0
320 IF JO=2 AND CH>6 THEN S(2)=S(2)+CAS(
    CH+(6-CH)*2+1):CAS(CH+(6-CH)*2+1)=0
330 FOR I=1 TO 12:SOM=SOM+CAS(I):NEXT I
340 IF SOM=1 AND JO=1 THEN S(1)=S(1)+1:GOTO
    620
350 IF SOM=1 AND JO=2 THEN S(2)=S(2)+1:GOTO
    620
360 FOR I=1 TO 6
370     SOM1=SOM1+CAS(I)
380     SOM2=SOM2+CAS(I+6)
390 NEXT I
400 IF (SOM1=0 AND JO=2)OR(SOM2=0 AND JO=1)
    THEN GOSUB 720
410 SOM=0:SOM1=0:SOM2=0
420 IF S(1)>24 OR S(2)>24 THEN GOTO 620
430 IF JO=1 THEN JO=2 ELSE JO=1
440 GOSUB 500
450 T1=TIMER
460 IF JO=2 THEN GOSUB 760
470 IF JO=1 AND CHO=1 THEN GOSUB 870
480 IF JO=1 AND CHO=2 THEN GOSUB 820
490 GOTO 190
500 '-----Affichage-----
510 PRINT "temps: ";USING "#####.##";(TIMER-
    T1)
520 PRINT
530 I=1:WHILE NOT BAB(I,1)=99
540     D$=INKEY$:IF D$<>" "THEN 580
550     FOR II=12 TO 7 STEP-1:PRINT
        BAB(I,II);NEXT II:PRINT
560     FOR II=1 TO 6:PRINT BAB(I,II);NEXT
        II:PRINT
570     PRINT STRING$(60,"-"):I=I+1:WEND
580     FOR II=12 TO 7 STEP-1:PRINT
        CAS(II);NEXT II:PRINT" <-joueur
        2","sc:"S(2)
590     FOR II=1 TO 6:PRINT CAS(II);NEXT
        II:PRINT" <-joueur 1","sc:"S(1)
600     PRINT "Le joueur"JO;"joue"
610     RETURN
620 '-----Fin de Partie-----
630     GOSUB 500
640     PRINT :PRINT,"Score: 1->"S(1)"/"S(2)"<-
        2"
650     IF S(1)=S(2) THEN PRINT :PRINT "Match
        nul":PRINT :GOTO 680
660     IF S(1)>S(2) THEN JO=1 ELSE JO=2
670     PRINT :PRINT "Le joueur"JO"gagne. Bravo
        !!":PRINT
680     PRINT "tp total:";USING"#.##";(TIMER-
        T0)/60
690     PRINT "Nombre total de coups
        joues:";NOMB
700     PRINT :PRINT ", "Au revoir et a bientot
        ...":PRINT :PRINT :END
710     END
720 '-----Peut pas jouer-----
730     IF JO=1 THEN JO=2 ELSE JO=1
740     PRINT"Le joueur"JO"ne peut pas jouer."
750     RETURN
760 '-----Joueur humain 2-----
770     'CH=INT(RND*6)+7:GOTO 570
780     INPUT "Case choisie par 2:";CH
790     IF CH<7 OR CH>12 THEN PRINT"Vous devez
        choisir un nombre de 7 a 12":GOTO 770
800     IF CAS(CH)=0 THEN PRINT "Cette case est
        vide !!":GOTO 770
810     RETURN
820 '-----Joueur humain 1-----
830     INPUT "Case choisie par 1:";CH
840     IF CH<1 OR CH>6 THEN PRINT"Vous devez
        choisir un nombre de 1 a 6":GOTO 830
850     IF CAS(CH)=0 THEN PRINT "Cette case est
        vide !!":GOTO 830
860     RETURN
870 '-----Ordinateur-----
880     T1=TIMER
890     Z(1,0)=0:FOR I=1 TO 12:Z(1,I)=CAS(I):
        NEXT I
900     II=0:FOR I=1 TO 6:IF CAS(I)=0 THEN
        II=II+1
910     NEXT I:IF II=5 THEN GOSUB 1460:GOTO 940
920     GOSUB 1200
930     GOSUB 960
940     PRINT "Case";CH
950     RETURN
960 '-----Quel coup jouer-----
970     J=1
980     FOR NIV=1 TO 3
990     IF J=2 THEN J=1 ELSE J=2
1000    FOR DEP=1 TO 1296 STEP 6^NIV
1010    FOR I=1 TO 5
1020    IF Z(DEP+I*6^(NIV-1),0)=99 THEN GOTO
        1050
1030    IF Z(DEP,0)=99 THEN SWAP Z(DEP,0),Z(DEP
        +I*6^(NIV-1),0):GOTO 1050
1040    IF J=1 THEN GOSUB 1160 ELSE GOSUB 1180
1050    NEXT I:NEXT DEP:NEXT NIV
1060    FOR I=1 TO 1296 STEP 6^3
1070    IF Z(I,0)=99 THEN GOTO 1140
1080    FOR J=1 TO 1296 STEP 6^3
1090    IF Z(J,0)=99 THEN GOTO 1110
1100    IF Z(I,0)<Z(J,0)THEN SW2=1
1110    NEXT J
1120    IF SW2=1 THEN SW2=0:GOTO 1140
1130    CH=(I-1)/6^3+1:RETURN
1140    NEXT I
1150    PRINT "ch..... et m..... ca d.....
        desole.":END
1160    IF Z(DEP,0)<Z(DEP+I*6^(NIV-1),0) THEN

```

```

      SWAP Z(DEP,0),Z(DEP+I*6^(NIV-1),0)
1170 RETURN
1180 IF Z(DEP,0)>Z(DEP+I*6^(NIV-1),0) THEN
      SWAP Z(DEP,0),Z(DEP+I*6^(NIV-1),0)
1190 RETURN
1200 '-----On fait l'arbre de choix-----
1210 J=1
1220 FOR NIV=1 TO 4
1230 FOR DEP=1 TO 6^4 STEP 6^(5-NIV)
1240 FOR I=0 TO 12:DE(I)=Z(DEP,I):NEXT I
1250 FOR CA=1 TO 6
1260 FOR I=0 TO 12:A(I)=DE(I):NEXT I
1270 IF A(0)=99 THEN GOTO 1390
1280 IF J=1 THEN C=CA ELSE C=CA+6
1290 IF A(C)=0 THEN FOR I=0 TO 12:A(I)=99:
      NEXT I:RIEN=RIEN+1:GOTO 1390
1300 B=A(C):A(C)=0
1310 FOR I=1 TO B
1320 C=C+1:IF C=13 THEN C=1
1330 A(C)=A(C)+1
1340 NEXT I
1350 SW=SW+1:IF SW=1000 THEN FOR I=0 TO 12:
      A(I)=99:NEXT:GOTO 1390
1360 IF A(C)>1 THEN 1300
1370 IF (J=1 AND C<7) THEN A(0)=A(0)+A(C+2*
      (6-C)+1):A(C+2*(6-C)+1)=0
1380 IF (J=2 AND C>6) THEN A(0)=A(0)-
      A(C+2*(6-C)+1):A(C+2*(6-C)+1)=0
1390 FOR I=0 TO 12
1400 Z(DEP+(CA-1)*6^(4-NIV),I)=A(I)
1410 IF RIEN=6 THEN Z(DEP+(CA-1)*6^(4-
      NIV),I)=DE(I)
1420 NEXT I:SW=0
1430 NEXT CA:RIEN=0:NEXT DEP
1440 IF J=1 THEN J=2 ELSE J=1
1450 NEXT NIV:RETURN
1460 '---Une seule case possible-----
1470 FOR I=1 TO 6
1480 IF CAS(I)>0 THEN CH=I
1490 NEXT I:RETURN
1500 '---Ecran du debut-----
1510 PRINT "
      Jeu de l'
      AWELE (version)"
1520 PRINT:PRINT
1530 PRINT" Le jeu de l'Awele est un jeu
      africain tres ancien. Cette version"
1540 PRINT"est connue au Ghana, par exemple.
      Il existe des centaines d'autres"
1550 PRINT"versions de ce jeu."
1560 PRINT
1570 PRINT "Regles de jeu:"
1580 PRINT "L' Awele se joue a deux joueurs.
      Le tableau du jeu comporte deux"
1590 PRINT "camps de 6 cases, un pour chaque
      joueur. Les joueurs jouent tour a"
1600 PRINT "tour. Le joueur doit choisir une
      case de son camp non vide."
1610 PRINT "Il prend les billes de cette
      case et les distribuent une a une"
1620 PRINT "dans les cases qui suivent (en
      sens inverse des aiguilles d'une"
1630 PRINT "montre). S'il pose la derniere
      bille dans une case non vide, il"
1640 PRINT "prend les billes de cette case,
      et recommence , jusqu'a ce qu'il"
1650 PRINT "qu'il tombe dans une case vide.
      Si cette case est dans son camp,"
1660 PRINT "il prend les billes de la case
      en face de la sienne. Si cette case"
1670 PRINT "est dans le camp adverse, il ne
      prend rien. Apres cela, c'est son"
1680 PRINT "adversaire qui joue, et ainsi de
      suite."
1690 PRINT "Le but du jeu est de gagner le
      plus de billes. Le jeu s'arrete"
1700 PRINT "s'il n'y a plus de billes sur le
      plateau, ou si l'un des joueurs"
1710 PRINT "a amasse plus de la moitie du
      total des billes."
1720 PRINT "
      <Tapez sur une
      touche pour jouer>"
1730 D$=INKEY$:IF D$="" THEN 1730
1740 CLS
1750 PRINT "Utilisation du programme:":PRINT
1760 PRINT "
      - Si vous jouez a deux, le
      joueur 1 choisit des cases de 1 a 6"
1770 PRINT "et le joueur 2 choisit des cases
      de 7 a 12"
1780 PRINT
1790 PRINT "
      - Si vous jouez contre
      l'ordinateur, vous etes le joueur 2"
1800 PRINT "vous pouvez selectionner les
      cases de 7 a 12 comme indique:"
1810 PRINT
1820 PRINT "
      12 11 10 9 8 7"
1830 PRINT "
      1-----1"
1840 PRINT "
      1 1 1 1 1 1 1
      <- Camp du joueur 2"
1850 PRINT "
      1-----1"
1860 PRINT "
      1 1 1 1 1 1 1
      <- Camp du joueur 1 ou ordinateur"
1870 PRINT "
      1-----1"
1880 PRINT "
      1 2 3 4 5
      6":PRINT
1890 PRINT " Les nombres dans les cases
      renseignent sur le nombre de billes"
1900 PRINT "contenues dans ces cases.":PRINT
1910 PRINT " Vous pouvez allez directement
      au tableau final en appuyant"
1920 PRINT "sur <ESPACE> lors du defilement
      des tableaux intermediaires."
1930 PRINT
1940 PRINT "
      <Tapez sur une
      touche pour jouer>"
1950 D$=INKEY$:IF D$="" THEN 1950
1960 RETURN

```

*Si vous voulez une copie du programme, envoyez quelques dizaines de francs (disquette et frais de port) à l'adresse suivante : COSMAO Jérémie, 591 rue Avril Coeuret, 76230 ISNEAUVILLE, tél : 35 61 19 94*

Ce programme suggère une possibilité de "technique pour gagner". Il s'agit que l'adversaire gagne le moins de billes. D'où l'idée d'étudier l'évolution et la formation des plus gros amas, ainsi que l'exposition à la prise par l'adversaire. Idée que nous n'avons pas encore mise à l'étude ...

[NDLR : pour en savoir plus sur l'Awélé ou sur les Awélés, le livre qu'il ne faut pas rater est celui de **Deledicq et Popova, Wari et Solo**, éditions CEDIC-Nathan, 1980.]

[NDLR : les élèves ont mis au point un programme offrant diverses possibilités de calcul sur le jeu d'Awele ; il est prévu pour Casio fx-7800 GC et fx-8800 GC. *Si vous voulez une copie de ce programme, adressez-vous à : Pierre Gélard, 52 allée des Halliers, 76230 Bois-Guillaume ]*